Problem:
Score : 600 points

Problem Statement

There are N points in a two-dimensional plane. The initial coordinates of the i-th point are (x_i, y_i). Now, each point starts moving at a speed of 1 per second, in a direction parallel to the x- or y- axis. You are given a character d_i that represents the specific direction in which the i-th point moves, as follows:

If d_i = R, the i-th point moves in the positive x direction;

If d_i = L, the i-th point moves in the negative x direction;

If d_i = U, the i-th point moves in the positive y direction;

If d_i = D, the i-th point moves in the negative y direction.

You can stop all the points at some moment of your choice after they start moving (including the moment they start moving).
Then, let x_{max} and x_{min} be the maximum and minimum among the x-coordinates of the N points, respectively. Similarly, let y_{max} and y_{min} be the maximum and minimum among the y-coordinates of the N points, respectively.

Find the minimum possible value of (x_{max} - x_{min}) \times (y_{max} - y_{min}) and print it.

Constraints

1 \leq N \leq 10^5

-10^8 \leq x_i,\ y_i \leq 10^8

x_i and y_i are integers.

d_i is R, L, U, or D.

Input

Input is given from Standard Input in the following format:

N
x_1 y_1 d_1
x_2 y_2 d_2
.
.
.
x_N y_N d_N

Output

Print the minimum possible value of (x_{max} - x_{min}) \times (y_{max} - y_{min}).

The output will be considered correct when its absolute or relative error from the judge's output is at most 10^{-9}.

Sample Input 1

2
0 3 D
3 0 L

Sample Output 1

0

After three seconds, the two points will meet at the origin. The value in question will be 0 at that moment.

Sample Input 2

5
-7 -10 U
7 -6 U
-8 7 D
-3 3 D
0 -6 R

Sample Output 2

97.5

The answer may not be an integer.

Sample Input 3

20
6 -10 R
-4 -9 U
9 6 D
-3 -2 R
0 7 D
4 5 D
10 -10 U
-1 -8 U
10 -6 D
8 -5 U
6 4 D
0 3 D
7 9 R
9 -4 R
3 10 D
1 9 U
1 -6 U
9 -8 R
6 7 D
7 -3 D

Sample Output 3

273

Solution in Fortran (gfortran v4.8.4) (judge verdict: Accepted)

module mod_bounding_box
  implicit none
  real(8), parameter :: infty = 1.e12
  type :: bounding_box
    real(8) :: l(-1:1), r(-1:1)
  end type bounding_box
  private
  public :: infty
  public :: mins, maxs
  public :: bounding_box, init, add, get, event
contains
  subroutine mins(x,y)
    implicit none
    real(8), intent(inout) :: x
    real(8), intent(in) :: y
    x = min(x,y)
    return
  end subroutine mins
  subroutine maxs(x,y)
    implicit none
    real(8), intent(inout) :: x
    real(8), intent(in) :: y
    x = max(x,y)
    return
  end subroutine maxs
  subroutine init(bb)
    implicit none
    type(bounding_box), intent(inout) :: bb
    bb%l = infty
    bb%r = -infty
    return
  end subroutine init
  subroutine add(bb,z,i)
    implicit none
    type(bounding_box), intent(inout) :: bb
    real(8), intent(in) :: z
    integer :: i
    call mins(bb%l(i),z)
    call maxs(bb%r(i),z)
    return
  end subroutine add
  function get(bb,t) result(d)
    implicit none
    type(bounding_box), intent(in) :: bb
    real(8), intent(in) :: t
    real(8) :: d, l, r
    integer :: i
    l = infty
    r = -infty
    do i = -1, 1
      call mins(l,bb%l(i)+real(i,8)*t)
      call maxs(r,bb%r(i)+real(i,8)*t)
    end do
    d = r-l
    return
  end function get
  function event(bb) result(t)
    implicit none
    type(bounding_box), intent(in) :: bb
    real(8) :: t(6)
    t(1) = bb%l(-1)-bb%l(0)
    t(2) = 0.5d0*(bb%l(-1)-bb%l(1))
    t(3) = bb%l(0)-bb%l(1)
    t(4) = bb%r(-1)-bb%r(0)
    t(5) = 0.5d0*(bb%r(-1)-bb%r(1))
    t(6) = bb%r(0)-bb%r(1)
    return
  end function event
end module mod_bounding_box
program minimum_bounding_box
  use mod_bounding_box
  implicit none
  integer :: n, i
  real(8) :: x, y, ans, events(14)
  type(bounding_box) :: xbb, ybb
  character(1) :: d
  character(64) :: s
  call init(xbb)
  call init(ybb)
  events = 0.d0
  read(*,*) n
  do i = 1, n
    read(*,*) x, y, d
    select case(d)
    case("R")
      call add(xbb,x,1)
      call add(ybb,y,0)
    case("L")
      call add(xbb,x,-1)
      call add(ybb,y,0)
    case("U")
      call add(xbb,x,0)
      call add(ybb,y,1)
    case("D")
      call add(xbb,x,0)
      call add(ybb,y,-1)
    end select
  end do
  events(1:6) = event(xbb)
  events(7:12) = event(ybb)
  events(13) = 0.d0
  events(14) = infty
  ans = infty**3
  do i = 1, 14
    if (events(i).lt.0.d0) cycle
    call mins(ans,get(xbb,events(i))*get(ybb,events(i)))
  end do
  write(s,'(f40.16)') ans
  write(*,'(a)') trim(adjustl(s))
  stop
end program minimum_bounding_box